
The smallest free number

Introduction

Consider the problem of computing the smallest natural number not in a
given finite set X of natural numbers. The problem is a simplification of a
common programming task in which the numbers name objects and X is
the set of objects currently in use. The task is to find some object not in
use, say the one with the smallest name.

The solution to the problem depends, of course, on how X is represented.
If X is given as a list without duplicates and in increasing order, then the
solution is straightforward: simply look for the first gap. But suppose X is
given as a list of distinct numbers in no particular order. For example,

[08, 23, 09, 00, 12, 11, 01, 10, 13, 07, 41, 04, 14, 21, 05, 17, 03, 19, 02, 06]

How would you find the smallest number not in this list?
It is not immediately clear that there is a linear-time solution to the

problem; after all, sorting an arbitrary list of numbers cannot be done in
linear time. Nevertheless, linear-time solutions do exist and the aim of this
pearl is to describe two of them: one is based on Haskell arrays and the other
on divide and conquer.

An array-based solution

The problem can be specified as a function minfree, defined by

minfree :: [Nat ]→ Nat
minfree xs = head ([0 .. ] \\ xs)

The expression us \\ vs denotes the list of those elements of us that remain
after removing any elements in vs:

(\\) :: Eq a ⇒ [a]→ [a]→ [a]
us \\ vs = filter (�∈ vs) us

2 Pearls of Functional Algorithm Design

The function minfree is executable but requires Θ(n2) steps on a list of
length n in the worst case. For example, evaluating minfree [n−1,n−2 .. 0]
requires evaluating i /∈ [n−1,n−2 .. 0] for 0 ≤ i ≤ n, and thus n(n + 1)/2
equality tests.

The key fact for both the array-based and divide and conquer solutions
is that not every number in the range [0 .. length xs] can be in xs. Thus the
smallest number not in xs is the smallest number not in filter (≤ n)xs, where
n = length xs. The array-based program uses this fact to build a checklist
of those numbers present in filter (≤ n) xs. The checklist is a Boolean array
with n +1 slots, numbered from 0 to n, whose initial entries are everywhere
False. For each element x in xs and provided x ≤ n we set the array element
at position x to True. The smallest free number is then found as the position
of the first False entry. Thus, minfree = search · checklist , where

search :: Array Int Bool → Int
search = length · takeWhile id · elems

The function search takes an array of Booleans, converts the array into a list
of Booleans and returns the length of the longest initial segment consisting
of True entries. This number will be the position of the first False entry.

One way to implement checklist in linear time is to use the function
accumArray in the Haskell library Data.Array . This function has the rather
daunting type

Ix i ⇒ (e → v → e) → e → (i , i)→ [(i , v)]→ Array i e

The type constraint Ix i restricts i to be an Index type, such as Int or
Char , for naming the indices or positions of the array. The first argument is
an “accumulating” function for transforming array entries and values into
new entries, the second argument is an initial entry for each index, the third
argument is a pair naming the lower and upper indices and the fourth is
an association list of index–value pairs. The function accumArray builds
an array by processing the association list from left to right, combining
entries and values into new entries using the accumulating function. This
process takes linear time in the length of the association list, assuming the
accumulating function takes constant time.

The function checklist is defined as an instance of accumArray :

checklist :: [Int ] → Array Int Bool
checklist xs = accumArray (∨) False (0,n)

(zip (filter (≤ n) xs) (repeat True))
where n = length xs

The smallest free number 3

This implementation does not require the elements of xs to be distinct, but
does require them to be natural numbers.

It is worth mentioning that accumArray can be used to sort a list of
numbers in linear time, provided the elements of the list all lie in some
known range (0,n). We replace checklist by countlist , where

countlist :: [Int ] → Array Int Int
countlist xs = accumArray (+) 0 (0,n) (zip xs (repeat 1))

Then sort xs = concat [replicate k x | (x , k)← countlist xs]. In fact, if we use
countlist instead of checklist , then we can implement minfree as the position
of the first 0 entry.

The above implementation builds the array in one go using a clever library
function. A more prosaic way to implement checklist is to tick off entries step
by step using a constant-time update operation. This is possible in Haskell
only if the necessary array operations are executed in a suitable monad, such
as the state monad. The following program for checklist makes use of the
library Data.Array .ST :

checklist xs = runSTArray (do
{a ← newArray (0,n) False;
sequence [writeArray a x True | x ← xs, x ≤ n];
return a})

where n = length xs

This solution would not satisfy the pure functional programmer because it
is essentially a procedural program in functional clothing.

A divide and conquer solution

Now we turn to a divide and conquer solution to the problem. The idea is
to express minfree (xs ++ys) in terms of minfree xs and minfree ys. We begin
by recording the following properties of \\:

(as ++ bs) \\ cs = (as \\ cs) ++ (bs \\ cs)
as \\ (bs ++ cs) = (as \\ bs) \\ cs
(as \\ bs) \\ cs = (as \\ cs) \\ bs

These properties reflect similar laws about sets in which set union ∪ replaces
++ and set difference \ replaces \\. Suppose now that as and vs are disjoint,
meaning as \\ vs = as, and that bs and us are also disjoint, so bs \\ us = bs.
It follows from these properties of ++ and \\ that

(as ++ bs) \\ (us ++ vs) = (as \\ us) ++ (bs \\ vs)

4 Pearls of Functional Algorithm Design

Now, choose any natural number b and let as = [0 .. b−1] and bs = [b..].
Furthermore, let us = filter (< b) xs and vs = filter (≥ b) xs. Then as and
vs are disjoint, and so are bs and us. Hence

[0 .. ] \\ xs = ([0 .. b−1] \\ us) ++ ([b .. ] \\ vs)
where (us, vs) = partition (< b) xs

Haskell provides an efficient implementation of a function partition p that
partitions a list into those elements that satisfy p and those that do not.
Since

head (xs ++ ys) = if null xs then head ys else head xs

we obtain, still for any natural number b, that

minfree xs = if null ([0 .. b−1] \\ us)
then head ([b .. ] \\ vs)
else head ([0 .. ] \\ us)
where (us, vs) = partition (< b) xs

The next question is: can we implement the test null ([0 .. b−1] \\ us) more
efficiently than by direct computation, which takes quadratic time in the
length of us? Yes, the input is a list of distinct natural numbers, so is us.
And every element of us is less than b. Hence

null ([0 .. b−1] \\ us ≡ length us b

Note that the array-based solution did not depend on the assumption that
the given list did not contain duplicates, but it is a crucial one for an efficient
divide and conquer solution.

Further inspection of the above code for minfree suggests that we should
generalise minfree to a function, minfrom say, defined by

minfrom :: Nat → [Nat ]→ Nat
minfrom a xs = head ([a .. ] \\ xs)

where every element of xs is assumed to be greater than or equal to a.
Then, provided b is chosen so that both length us and length vs are less than
length xs, the following recursive definition of minfree is well-founded:

minfree xs = minfrom 0 xs
minfrom a xs | null xs = a

| length us b − a = minfrom b vs
| otherwise = minfrom a us

where (us, vs) = partition (< b) xs

The smallest free number 5

It remains to choose b. Clearly, we want b > a. And we would also like to
choose b so that the maximum of the lengths of us and vs is as small as
possible. The right choice of b to satisfy these requirements is

b = a + 1 + n div 2

where n = length xs. If n �= 0 and length us < b − a, then

length us ≤ n div 2 < n

And, if length us = b − a, then

length vs = n − n div 2− 1 ≤ n div 2

With this choice the number of steps T (n) for evaluating minfrom 0 xs
when n = length xs satisfies T (n) = T (n div 2) + Θ(n), with the solution
T (n) = Θ(n).

As a final optimisation we can avoid repeatedly computing length with a
simple data refinement, representing xs by a pair (length xs, xs). That leads
to the final program

minfree xs = minfrom 0 (length xs, xs)
minfrom a (n, xs) | n 0 = a

| m b − a = minfrom b (n −m, vs)
| otherwise = minfrom a (m, us)

where (us, vs) = partition (< b) xs
b = a + 1 + n div 2
m = length us

It turns out that the above program is about twice as fast as the incremental
array-based program, and about 20% faster than the one using accumArray .

Final remarks

This was a simple problem with at least two simple solutions. The second
solution was based on a common technique of algorithm design, namely
divide and conquer. The idea of partitioning a list into those elements less
than a given value, and the rest, arises in a number of algorithms, most
notably Quicksort. When seeking a Θ(n) algorithm involving a list of n
elements, it is tempting to head at once for a method that processes each
element of the list in constant time, or at least in amortized constant time.
But a recursive process that performs Θ(n) processing steps in order to
reduce the problem to another instance of at most half the size is also good
enough.

6 Pearls of Functional Algorithm Design

One of the differences between a pure functional algorithm designer and
a procedural one is that the former does not assume the existence of arrays
with a constant-time update operation, at least not without a certain amount
of plumbing. For a pure functional programmer, an update operation takes
logarithmic time in the size of the array.1 That explains why there sometimes
seems to be a logarithmic gap between the best functional and procedural
solutions to a problem. But sometimes, as here, the gap vanishes on a closer
inspection.

1 To be fair, procedural programmers also appreciate that constant-time indexing and updating
are only possible when the arrays are small.

A surpassing problem

Introduction

In this pearl we solve a small programming exercise of Martin Rem (1988a).
While Rem’s solution uses binary search, our solution is another application
of divide and conquer. By definition, a surpasser of an element of an array
is a greater element to the right, so x [j ] is a surpasser of x [i ] if i < j
and x [i ] < x [j ]. The surpasser count of an element is the number of its
surpassers. For example, the surpasser counts of the letters of the string
GENERATING are given by

G E N E R A T I N G
5 6 2 5 1 4 0 1 0 0

The maximum surpasser count is six. The first occurrence of letter E has six
surpassers, namely N, R, T, I, N and G. Rem’s problem is to compute the
maximum surpasser count of an array of length n > 1 and to do so with an
O(n log n) algorithm.

Specification

We will suppose that the input is given as a list rather than an array. The
function msc (short for maximum surpasser count) is specified by

msc :: Ord a ⇒ [a]→ Int
msc xs = maximum [scount z zs | z : zs ← tails xs]
scount x xs = length (filter (x <) xs)

The value of scount x xs is the surpasser count of x in the list xs and tails
returns the nonempty tails of a nonempty list in decreasing order of length:1

tails [ ] = [ ]
tails (x : xs) = (x : xs) : tails xs

The definition of msc is executable but takes quadratic time.
1 Unlike the standard Haskell function of the same name, which returns the possibly empty tails

of a possibly empty list.

8 Pearls of Functional Algorithm Design

Divide and conquer

Given the target complexity of O(n log n) steps, it seems reasonable to head
for a divide and conquer algorithm. If we can find a function join so that

msc (xs ++ ys) = join (msc xs) (msc ys)

and join can be computed in linear time, then the time complexity T (n) of
the divide and conquer algorithm for computing msc on a list of length n
satisfies T (n) = 2T (n/2)+O(n), with solution T (n) = O(n log n). But it is
fairly obvious that no such join can exist: too little information is provided
by the single number msc xs for any such decomposition.

The minimal generalisation is to start out with the table of all surpasser
counts:

table xs = [(z , scount z zs) | z : zs ← tails xs]

Then msc = maximum ·map snd · table. Can we find a linear-time join to
satisfy

table (xs ++ ys) = join (table xs) (table ys)

Well, let us see. We will need the following divide and conquer property of
tails:

tails (xs ++ ys) = map (++ys) (tails xs) ++ tails ys

The calculation goes as follows:

table (xs ++ ys)

= {definition}
[(z , scount z zs) | z : zs ← tails (xs ++ ys)]

= {divide and conquer property of tails}
[(z , scount z zs) | z : zs ←map (++ys) (tails xs) ++ tails ys]

= {distributing ← over ++}
[(z , scount z (zs ++ ys)) | z : zs ← tails xs] ++
[(z , scount z zs) | z : zs ← tails ys])

= {since scount z (zs ++ ys) = scount z zs + scount z ys}
[(z , scount z zs + scount z ys) | z : zs ← tails xs] ++
[(z , scount z zs) | z : zs ← tails ys])

= {definition of table and ys = map fst (table ys)}
[(z , c + scount z (map fst (table ys))) | (z , c)← table xs] ++ table ys

A surpassing problem 9

Hence join can be defined by

join txs tys = [(z , c + tcount z tys) | (z , c)← txs] ++ tys
tcount z tys = scount z (map fst tys)

The problem with this definition, however, is that join txs tys does not take
linear time in the length of txs and tys.

We could improve the computation of tcount if tys were sorted in
ascending order of first component. Then we can reason:

tcount z tys

= {definition of tcount and scount}
length (filter (z <) (map fst tys))

= {since filter p ·map f = map f · filter (p · f )}
length (map fst (filter ((z <) · fst) tys))

= {since length ·map f = length}
length (filter ((z <) · fst) tys)

= {since tys is sorted on first argument}
length (dropWhile ((z ≥) · fst) tys)

Hence

tcount z tys = length (dropWhile((z ≥) · fst) tys) (2.1)

This calculation suggests it would be sensible to maintain table in ascending
order of first component:

table xs = sort [(z , scount z zs) | z : zs ← tails xs]

Repeating the calculation above, but for the sorted version of table, we find
that

join txs tys = [(x , c + tcount x tys) | (x , c)← txs] ∧∧ tys (2.2)

where ∧∧ merges two sorted lists. Using this identity we can now calculate
a more efficient recursive definition of join. One of the base cases, namely
join [ ] tys = tys, is immediate. The other base case, join txs [ ] = txs, follows
because tcount x [ ] = 0. For the recursive case we simplify

join txs@((x , c) : txs ′) tys@((y , d) : tys ′) (2.3)

by comparing x and y . (In Haskell, the @ sign introduces a synonym, so txs
is a synonym for (x , c) : txs ′; similarly for tys.) Using (2.2), (2.3) reduces to

((x , c + tcount x tys) : [(x , c + tcount x tys) | (x , c)← txs ′]) ∧∧ tys

10 Pearls of Functional Algorithm Design

To see which element is produced first by ∧∧ we need to compare x and y . If
x < y , then it is the element on the left and, since tcount x tys = length tys
by (2.1), expression (2.3) reduces to

(x , c + length tys) : join txs ′ tys

If x = y , we need to compare c + tcount x tys and d . But d = tcount x tys ′

by the definition of table and tcount x tys = tcount x tys ′ by (2.1), so (2.3)
reduces to (y , d) : join txs tys ′. This is also the result in the final case x > y .

Putting these results together, and introducing length tys as an additional
argument to join in order to avoid repeating length calculations, we arrive
at the following divide and conquer algorithm for table:

table [x ] = [(x , 0)]
table xs = join (m − n) (table ys) (table zs)

where m = length xs
n = m div 2
(ys, zs) = splitAt n xs

join 0 txs [ ] = txs
join n [ ] tys = tys
join n txs@((x , c) : txs ′) tys@((y , d) : tys ′)

| x < y = (x , c + n) : join n txs ′ tys
| x ≥ y = (y , d) : join (n−1) txs tys ′

Since join takes linear time, table is computed in O(n log n) steps, and so
is msc.

Final remarks

It is not possible to do better than an O(n log n) algorithm for computing
table. The reason is that if xs is a list of distinct elements, then table xs pro-
vides enough information to determine the permutation of xs that sorts xs.
Moreover, no further comparisons between list elements are required. In fact,
table xs is related to the inversion table of a permutation of n elements;
see Knuth (1998): table xs is just the inversion table of reverse xs. Since
comparison-based sorting of n elements requires Ω(n log n) steps, so does
the computation of table.

As we said in the Introduction for this pearl, the solution in Rem (1998b)
is different, in that it is based on an iterative algorithm and uses binary
search. A procedural programmer could also head for a divide and conquer
algorithm, but would probably prefer an in-place array-based algorithm
simply because it consumes less space.

